

TSS端末としてのマイコン利用法(3) : BASICでかかれた英語清書システムROFFのサブシステム

桑折, 範彦
九州大学工学部応用原子核工学教室

<https://doi.org/10.15017/1468077>

出版情報 : 九州大学大型計算機センター広報. 16 (4), pp. 325-334, 1983-07-25. 九州大学大型計算機センター
バージョン :
権利関係 :

TSS 端末としてのマイコン利用法(3) — BASIC で書かれた英文清書システム ROFF のサブシステム —

桑折 範彦*

1. はじめに

英語で論文を作成しなければならない我々研究者にとって、英文清書システムの必要性は高い。九大大型計算機センター（以後センターと書く）においては、ROFF [1] , ATF [2] のシステムが使用出来る。両システム共機能が豊富で便利であるが、論文を修正するたびにセンターまで出力を取りに行かなければならないのが一つの難点である。TSS 端末としてマイコンを利用している場合、論文作成、修正時にはマイコンだけでテキストファイルを作成し、また、清書を出力できて、最終的にはそのテキストファイルをセンターの計算機に転送して、センターの優れたシステムを利用することが出来ると便利であると考えられる。マイコンで出来ることをマイコンで行なうことによって、マイコンだけでなく大型計算機もそれを必要とする仕事に有効に利用することが可能になる。そこで私自身が良く利用している ROFF** について、そのコマンドの一部分必要なもののみを解読して清書する ROFF のサブシステム "SROFF" を BASIC で作成したので、ここに紹介する。

ここでは、ROFF システム及びそのコマンドについては文献 [1] を参照してもらおうか、あるいは、すでに知っているかと仮定する。SROFF における制限、使用法、プログラムの順に説明する。

2. SROFF における制限

SROFF では先に述べた様に、ROFF のコマンドの中で一部必要なものだけを解読する。これは BASIC インタープリタによる実行速度を考慮して、ある程度の処理速度を確保するためと、プログラムを簡単にしたいためである。付録 1 に、SROFF で使用できるコマンドと制限を ROFF のコマンド一覧表に加えて示した。

SROFF が解読するコマンドは以下に示すものだけで他は無視される。空白行の挿入：. LS , . SP ; 行の出力：. BR , . TB , . IN , . CL ; 頁の変更：. BP ; 行の詰め：. FI , . FI N ; センタリング：. CE , . CE N ; 制御文字定義：. CC *S , . CC *E , . CC *M ; マクロ定義：. MV. 自動分節、辞書登録、サブタイトル指定等の機能は全て省いた。全てを盛り込もうとして、プログラムが完成出来なくなることを恐れたためである。

コマンドは、ROFF では大文字、小文字いずれも許されているが、SROFF では大文字だけに制限した。コマンドを大文字にするのは、ROFF の場合でもテキスト部分とコマンド部分を見分けやすくするのに役に立つ。

頁のレイアウトに関しては、1 頁の行数、1 行の文字数等の論文推敲上あまり重要でないパラメータを 1 段組、5 6 行/頁、6 0 文字/行とプログラム内で固定している。デフォルト値を変更すればある程度レイアウトを変えることが出来る様にプログラムしてある。頁番号も自動的に右肩に印字さ

* 九州大学工学部応用原子核工学教室

** 九大だけでなく東大等の大型計算機センターにも類似のシステムがあると聞いている。

れる様に決めた。

自動分節は行わないが、.FI が指定されていると行の終りとして空白だけでなくハイフンも区切として考慮される。また、右調整も、自動分節機能を省いたので、行なわない。

マクロ定義の個数は20ヶまでに制限されていて、それ以上の定義があった時は無視される。しかし、必要があれば後述するプログラムリスト内のDIM文中、MAC1\$(20),MAC2\$(20)とNMX=20の20を大きくすれば変更できる。定義されたものは清書後にリストされる。

その他、初期値等いくつかの点でROFFの処理と異っている所があり得るが、大きな問題は生じない。細かい点についてはプログラムリストを読んで理解していただきたい。

3. 使用法

プログラムはIF800のBASICで作成したが、標準的なマイクロソフト系のBASICで書かれているので多くのマイコンで2,3の変更だけで使用できると思われる(PC8800,9800では変更なしで動く)。また、Z80CPU用汎用OSであるCP/Mの下でのBASIC80(MBASIC)インタープリタでも2,3の修正で使用できる。普通のBASICの場合、CP/MのMBASICの場合に分けて簡単に使用法を説明する。

普通のBASICの場合には、プログラムリストに示されるSROFFプログラムをRUNさせると、テキストファイル名が要求され、フロッピーディスクにあるそのテキストファイルからからSROFF用のテキスト(コマンドを含む)を読んでくる。この場合のSROFF用テキストファイルはBASICのコメント文(REM文)の集合で、SROFFはBASICの文番号の部分は無視して、コメント文を示す▼(プライム)より後をテキストと解釈する。だから、テキストファイルは一般にスクリーンエディット可能なBASICで▼(プライム)で始まるコメント文だけのプログラムとして作成すれば良いので、簡単に作成、修正が出来る。ただし、テキストファイルは必ず、いわゆる、アスキーセーブする必要がある。テキストファイルはいくつか(最大6ヶ)に分割されていてもよく、ファイル名が要求された時指定すれば連続したファイルと見なされる。BASICでは大きなテキストファイルは1度にプログラム領域に入らないことが有るため、分割して作成出来る様にして、清書時に一連のものと見なされる様にした。

CP/MのMBASICの場合には、テキストファイルは一般にエディタ(EDIT80あるいはWORD MASTER等***)によって作成するので、BASICの文番号およびコメント文の印▼(プライム)のないテキストファイルの方が作り易い。CP/M版SROFFでは、このCP/Mのテキストファイル、前述の文番号付BASICコメント文のテキストファイルのいずれかが選択できる。下にCP/M版SROFFに対する入力例を示した、TEXT TYPE=の行以外は普通のBASICの場合に対応している。

*** CP/Mには、この他にWORDSTARというエディタ・ワードプロセッサがあり英文清書についても便利で良く出来ていると聞いている。

```

RUN
      ===== SROFF (SUB-SET OF ROFF) =====
THIS PROGRAM CAN PRINT (i) Formatted text controlled by SROFF-system,
                        or (ii) Original text file image.
FORMATTED TEXT PRINT (Y/N)? Y/
<< FORMATTED TEXT PRINT >>
TEXT TYPE= CPM.TEXT (0) or BASIC.REM (1)? 0/
ON SCREEN (0) or ON PRINTER (1)? 1/
FILE NAME=NBHP1.TXT/
FILE NAME=NBHP2.TXT/
FILE NAME=/
ADJUST PAGE HEAD OK (Y/N)? Y/
PRINT PAGE: PINI,PEND = ? 1,10/

```

これらのテキストファイル(センターのROFFを意識して作成したものは、センターに端末プログラムによって転送すればROFFに適応したデータセットとなる。端末プログラムには、フロッピーディスクにあるファイルを転送する機能が必要である。BASICのコメント文の形式のファイルは▼(プライム)を取ってセンターTSSのEDITモードの状態で行番号つきで転送すればよく、CP/Mでのテキストファイルは行番号がないからINPUTモードで1行ずつ転送すればよい。我々の研究室ではIFTSS,CPMTSS端末プログラム[3]によって転送を実現している。

4. プログラム

プログラムリストを付録2に示した。CP/MのMBASICに対しては、リスト中の注に従って変更すればよい。マイクロソフト系のBASICでもOPEN文の形式が異なるものがあるので、その形式を合わせる必要がある。FM8,7,11はこの類に入ると考えられるので、これらのマイコンでも使用可能であろう。プログラムの構造はコマンドの追加、変更が容易である様に簡単にしたつもりである。自動分節機能を除けば他のROFFコマンドを組込むことはそれ程難しくない。私自身はSROFFが論文の作成、推敲のために必要な最小限の機能を持っていると考えて、プログラムはここまでとして、論文を書く方に時間を使いたいと思っている。

なお、プログラムはセンターのデータセットとして保存しているので各端末より読み出すことが出来る。データセット名は▼F0783.SROFF.TEXT▼であり、行番号はBASICのプログラムの文番号に一致させてある。

5. おわりに

最近のマイコンは性能がよくなり、プリンタについても高分解能のものが入手できるようになってきた。今後、ワードプロセッサに類することは、直接オフセット印刷用の図形の同時編集など複雑なことがなければ、大型計算機を必ずしも必要としなくなっている様に感ずる。ここで紹介したSROFFにもセンターのROFFと同じ機能を持たせられるだろうし、ROFFとの互換性を無視すれば論文で良く使われるギリシャ文字の利用など便利な点を加えていくことが出来るだろう。ともあれ、現在のSROFFはセンターのROFFのサブシステムとして互換性が保たれている点に特徴がある。これからこのSROFFを使用される方は、マイコンのシステムに応じて変更を加え、独自の英文清書プログラムを作成されてはいかかと思う。

参考文献

1. 牛島，横山 「英文文書清書システム ROFF の改訂と使用について」 九大大型計算機センター広報 14，2，1981，242 - 261.
2. 武富，遠矢，影川 「英論文編集システム ATF の使用法」 九大大型計算機センター広報 15，1，1982，81 - 126.
3. 桑折，丸林，川村，酒井 「TSS 端末としてのマイコン利用法（2）-インテリジェント端末プログラムの処理速度の比較-」 九大大型計算機センター広報 16，3，1983，232 - 234.

付録1. ROFF コマンド一覧と SROFF における制限

種別	コマンド	パラメータ	SROFF での制限 (カッコ内は変数名)	
枠組指定	PL	n	—	56行/頁(LMAX)
	M1	n	—	
	M2	n	—	
	M3	n	—	
	M4	n	—	
	SC	n	—	1段組のみ, 60文字/行(CMAX)
	DC	i j k	—	
	HS	n	—	5文字(HDC)
サブタイトル指定	HE	S	—	サブタイトル関係無視
	OH	S	—	頁番号は順に右肩に出力される
	EH	S	—	
	FO	S	—	
	OF	S	—	
	EF	S	—	
行の詰め	FI	なし	○	初期モード(SFIL=0)
		N	○	
右調整	AD	なし	—	右端調整しない(自動分節しないため)
		N	—	
センタリング	CE	なし	○	初期モード(SCE=0)
		N	○	
ハイフン挿入	HY	なし	—	自動分節しない
		N	—	
例外辞書	EN	W	—	例外辞書なし
	DE	W	—	
空白挿入	LS	n	○	n=1(初期値, LSP)
	SP	n	○	n=0(初期値, SPC)
	SL	n	—	
	NE	n	—	
行の出力	BR	なし	○	
	TB	n	○	n=0(初期値, TAB)
	IN	n	○	n=0(初期値, CIN)
	CL	n w	○	
	RA	w	—	

解 説

種別	コマンド	パラメータ	SROFFでの制限 (カッコ内は変数名)	
ページの 変更	BP	なし	○	
	OP	なし	—	
	PA	n	—	
	SK	n	—	
制御用文 字の定義	CC	*S	○	* = " @ " (初期値, ZSPS)
		*P	—	
		*C	—	
		*E	○	* = " ¥ " (初期値, ZESS)
		*M	○	* = " \$ " (初期値, ZMAS)
		*H	—	
		*L	—	
		*B	—	
		*F	—	
なし	—			
マクロ定義	MV	d V d S d	○	20ケまで定義可
メッセージ	MS		—	
ポーズ機能	TP		—	
	PS	なし N	— —	

○印は SROFF で使用可。—印は無視される。

付録2. プログラムリスト "SROFF"

```

10 '===== SROFF =====
20 ' SUB SET OF "ROFF" OF COMPUTER CENTER, KYUSHU UNIV.
30 ' PROGRAMMED BY N.KOORI (1982.NOV.15) <FOR IF800 or PC8800>
40 ' *** MODIFICATION FOR CP/M IS INCLUDED (INDICATED BY ***)
50 ' modified by N.KOORI (1983.APR.25)
60 '=====
70 DEFINIT A-Z:DIM MAC1$(20),MAC2$(20),FILEIN$(5)
80 '----- DEFAULT VALUES -----
90 CMAX=60:LMAX=56:HDC=5: ZSP$="0":ZES$="¥":ZMA$=" $" '## DEFAULTS
100 PINI=1:PEND=999: NM=0:NMX=20 '## DEFAULTS
110 SFIL=0:SCE=0:SPR=0:SCL=0:LSP=1:SPC=0:TAB=0:CIN=0:PAGE=1:LNO=0:EPL=0
120 '----- START PROGRAMME -----
130 PRINT CHR$(12):PRINT:PRINT
140 PRINT "          ===== SROFF (SUB-SET OF ROFF) ====="
150 PRINT "THIS PROGRAM CAN PRINT (i) Formatted text controlled by SROFF-system,
      ":PRINT SPACE$(20)+"or (ii) Original text file image."
160 LINE INPUT "FORMATTED TEXT PRINT (Y/N)? ";OK$:IF OK$<>"Y" AND OK$<>"y" THEN
      530
170 '----- FORMATTED TEXT PRINT -----
180 PRINT "<< FORMATTED TEXT PRINT >>"
190 ' INPUT "TEXT TYPE= CPM.TEXT (0) or BASIC.REM (1)";TTZ
      :IF TTZ<0 OR TTZ>1 THEN 190 '*** FOR CPM, INSERT THIS LINE.
200 INPUT "ON SCREEN (0) or ON PRINTER (1)";SPR:IF SPR<0 OR SPR>1 THEN SPR=0
210 IF SPR=0 THEN PINI=999
220 FOR FJJ=0 TO 5
230 LINE INPUT "FILE NAME=":FILEIN$(FJJ)
240 IF FILEIN$(FJJ)="" THEN MFJJ=FJJ-1:GOTO 270
250 NEXT FJJ
260 MFJJ=5
270 IF SPR=0 THEN 310
280 LPRINT CHR$(12);
290 LINE INPUT "ADJUST PAGE HEAD OK (Y/N)? ";OK$:IF OK$<>"Y" AND OK$<>"y" THEN
      290
300 INPUT "PRINT PAGE; PINI,PEND = ":PINI,PEND
      :IF PINI>0 AND PINI<=PEND THEN SPR=0 ELSE GOTO 300
310 FJJ=0: OPEN FILEIN$(FJJ) FOR INPUT AS #2 '----- OPEN FILE
320 ' FJJ=0:OPEN "I",2,FILEIN$(FJJ) '*** FOR CPM, CHANGE PREC.LINE BY THIS.
330 IF PAGE<=PINI AND PAGE<=PEND THEN SPR=1 ELSE SPR=0
340 IF SPR=1 THEN LPRINT SPACE$(HDC+CMAX+2):LPRINT USING "###";PAGE
      ELSE PRINT SPACE$(HDC+CMAX+2):PRINT USING "###";PAGE
350 '----- MAIN LOOP -----
360 IF EOF(2) GOTO 430
370 LINE INPUT #2,INTXT$
380 ' IF TTZ=0 THEN 400 '*** FOR CPM, INSERT THIS LINE.
390 SS=INSTR(INTXT$,""):IF SS<>0 AND SS<=7 THEN INTXT$=MID$(INTXT$,SS+1) ELSE
      PRINT "ERR IN TEXT FILE":PRINT INTXT$:GOTO 500
400 SS=INSTR(INTXT$,"."):IF SS=1 THEN GOSUB 600:GOTO 420 '==COMMAND LINE
410 GOSUB 1560:GOTO 420 '==TEXT LINE
420 GOTO 360
430 IF FJJ<MFJJ THEN CLOSE #2:FJJ=FJJ+1:OPEN FILEIN$(FJJ) FOR INPUT AS #2:GOTO
      360
440 ' IF FJJ<MFJJ THEN CLOSE #2:FJJ=FJJ+1:OPEN "I",2,FILEIN$(FJJ):GOTO 360 '***
      FOR CPM, CHANGE PREC.LINE BY THIS.
450 GOSUB 890
460 IF NM>0 THEN PRINT "==== MACRO TERM LIST ====="
470 FOR NZ=1 TO NM

```

```

480 PRINT NZ,MAC1$(NZ),MAC2$(NZ)
490 NEXT NZ
500 CLOSE #2: END
510 '=====
520 '
530 '----- ORIGINAL FILE IMAGE "PRINT" -----
540 PRINT "<< ORIGINAL FILE IMAGE >>"
550 INPUT "ON SCREEN (0) or ON PRINTER (1)";SPR:IF SPR<0 OR SPR>1 THEN SPR=0
560 LINE INPUT "FILE NAME=";FILEIN$
570 IF SPR=0 THEN 600
580 LPRINT CHR$(12);
590 LINE INPUT "ADJUST PAGE HEAD OK (Y/N)? ";OK$:IF OK$<>"Y" AND OK$<>"N" THEN
590
600 OPEN FILEIN$ FOR INPUT AS #2
610 ' OPEN "I",2,FILEIN$ '*** FOR CPM, CHANGE PREC.LINE BY THIS.
620 IF EOF(2) GOTO 660
630 LINE INPUT #2,INTXT$
640 IF SPR=1 THEN LPRINT INTXT$ ELSE PRINT INTXT$
650 GOTO 620
660 CLOSE #2: END
670 '=====
680 '----- ROFF COMMAND PROC. -----
690 NP=1 '## COMMAND CMD$=inter Periods
700 NPP=INSTR(NP+1,INTXT$,"."):IF NPP=0 GOTO 720
710 CMD$=MID$(INTXT$,NP+1,NPP-NP-1):NP=NPP+1:GOSUB 730:GOTO 700
720 CMD$=MID$(INTXT$,NP+1):GOSUB 730:RETURN
730 '----- EXEC ROFF COMMAND -----
740 GOSUB 1160 '## CMD PARA. CCC$,CCP1$,CCP2$
750 IF CCC$<>"FI" THEN 760 ELSE IF LEFT$(CCP1$,1)="N" THEN SFIL=1:GOTO 860 ELSE
SFIL=0:GOTO 860
760 IF CCC$<>"CE" THEN 770 ELSE IF LEFT$(CCP1$,1)="N" THEN SCE=1:GOTO 860 ELSE
SCE=0:GOTO 860
770 IF CCC$<>"BR" THEN 780 ELSE GOSUB 890:GOTO 860
780 IF CCC$<>"BP" THEN 790 ELSE GOSUB 970:GOTO 860
790 IF CCC$<>"LS" THEN 800 ELSE LSP=VAL(CCP1$):IF LSP=1 AND LSP<=3 THEN 860
ELSE LSP=1:GOTO 860
800 IF CCC$<>"SP" THEN 810 ELSE SPC=VAL(CCP1$):GOSUB 1010:GOTO 860
810 IF CCC$<>"TB" THEN 820 ELSE TAB=VAL(CCP1$):GOTO 860
820 IF CCC$<>"IN" THEN 830 ELSE CIN=VAL(CCP1$)-1:GOTO 860
830 IF CCC$<>"CC" THEN 840 ELSE CHT$=LEFT$(CCP1$,1):CHT2$=MID$(CCP1$,2,1): IF
CHT2$="S" THEN ZSP$=CHT$:GOTO 860 ELSE IF CHT2$="M" THEN ZMA$=CHT$:GOTO 860
ELSE IF CHT2$="E" THEN ZES$=CHT$:GOTO 860 ELSE GOTO 860
840 IF CCC$<>"MV" THEN 850 ELSE GOSUB 1350:GOTO 860
850 IF CCC$<>"CL" THEN 860 ELSE GOSUB 890:ATXT$=SPACE$(CMAX):
MID$(ATXT$,VAL(CCP1$),LEN(CCP2$))=CCP2$:SCL=1:GOTO 860
860 RETURN
870 '=====
880 '===== SUBROUTINES =====
890 '----- BREAK TEXT LINE i.e. PRINT 1-LINE PRECEDED -----
900 IF EPL=0 THEN IF SCL=0 THEN RETURN ELSE PTXT$=ATXT$:SCL=0:GOTO 930
910 LPTXT=LEN(PTXT$):IF LPTXT>CMAX THEN PTXT$=LEFT$(PTXT$,CMAX)
920 IF SCL=1 THEN MID$(PTXT$,1,TAB)=LEFT$(ATXT$,TAB):SCL=0 '>> INSERT
930 GOSUB 1310 '>> CHANGE SP-CHARA.
940 IF SPR=1 THEN LPRINT SPACE$(HDC)+PTXT$ ELSE PRINT SPACE$(HDC)+PTXT$
950 GOSUB 1420 '>> LINE & PAGE CONTROL
960 EPL=0:RETURN

```

```

970 '----- BREAK PAGE -----
980 GOSUB 890:IF LNO=0 THEN RETURN
990 GOSUB 1100 ')>> EJECT PAGE
1000 RETURN
1010 '----- SPACE LINE -----
1020 GOSUB 890 'IF LNO<>0 THEN SPC=SPC-1
1030 FOR I=1 TO SPC
1040 IF SPR=1 THEN LPRINT ELSE PRINT
1050 LNO=LNO+1
1060 IF LNO<LMAX THEN 1080
1070 GOSUB 1100 ')>> EJECT PAGE
1080 NEXT I
1090 RETURN
1100 '----- EJECT PAGE -----
1110 LNO=0: PAGE=PAGE+1
1120 IF SPR=1 THEN LPRINT CHR$(12);
1130 IF PAGE>=PINI AND PAGE<=PEND THEN SPR=1 ELSE SPR=0
1140 IF SPR=1 THEN LPRINT SPACE$(HDC+CMA*2);:LPRINT USING "###";PAGE
ELSE PRINT SPACE$(HDC+CMA*2);:PRINT USING "###";PAGE
1150 RETURN
1160 '----- CMD PARA. -----
1170 CCC$="":CCP1$="":CCP2$=""
1180 PSP=INSTR(COMD$," "):IF PSP=0 THEN CCC$=LEFT$(COMD$,2):GOTO 1300
1190 IF PSP=1 THEN COMD$=MID$(COMD$,2):GOTO 1180
1200 IF PSP<>3 THEN 1290
1210 CCC$=LEFT$(COMD$,2):IF LEN(COMD$)>=4 THEN COMD$=MID$(COMD$,4):GOTO 1220
ELSE 1300
1220 PSP=INSTR(COMD$," "):IF PSP=0 THEN CCP1$=COMD$:GOTO 1300
1230 IF PSP=1 THEN COMD$=MID$(COMD$,2):GOTO 1220
1240 IF CCC$<"MU" THEN 1250 ELSE CCP1$=COMD$:GOTO 1300
1250 CCP1$=LEFT$(COMD$,PSP-1):IF LEN(COMD$)>=PSP+1 THEN COMD$=MID$(COMD$,PSP+1):
GOTO 1260 ELSE 1300
1260 PSP=INSTR(COMD$," "):IF PSP=0 THEN CCP2$=COMD$:GOTO 1300
1270 IF PSP=1 THEN COMD$=MID$(COMD$,2):GOTO 1260
1280 CCP2$=LEFT$(COMD$,PSP-1)
1290 ')>> HERE, ERROR POINT
1300 RETURN
1310 '----- CHANGE SP-CHARA -----
1320 PSP=0
1330 PSP1=INSTR(PSP+1,PTXT$,ZSP$):IF PSP1=0 THEN RETURN
1340 IF PSP1=1 THEN MID$(PTXT$,PSP1,1)=" ":PSP=PSP1:GOTO 1330 ELSE IF MID$(PTXT$
,PSP1-1,1)=ZES$ THEN PTXT$=LEFT$(PTXT$,PSP1-2)+MID$(PTXT$,PSP1-1):GOTO
1330 ELSE MID$(PTXT$,PSP1,1)=" ":PSP=PSP1:GOTO 1330
1350 '----- MACRO DEF. -----
1360 IF NM>=NM* THEN RETURN
1370 NM=NM+1:DLC$=LEFT$(CCP1$,1)
1380 PDC=INSTR(2,CCP1$,DLC$):IF PDC=0 THEN NM=NM-1:RETURN
1390 MAC1$(NM)=MID$(CCP1$,2,PDC-2)
1400 PDC1=INSTR(PDC+1,CCP1$,DLC$): IF PDC1=0 THEN MAC2$(NM)=MID$(CCP1$,PDC+1):
RETURN ELSE MAC2$(NM)=MID$(CCP1$,PDC+1,PDC1-PDC-1):RETURN
1410 MAC2$(NM)=MID$(CCP1$,PDC+1,PDC1-PDC-1)
1420 '----- LINE & PAGE CONTROL -----
1430 IF LSP=1 THEN 1460
1440 IF LSP<>2 THEN 1450 ELSE IF SPR=1 THEN LPRINT:GOTO 1460 ELSE PRINT:GOTO
1460
1450 IF LSP<>3 THEN 1460 ELSE IF SPR=1 THEN LPRINT:LPRINT:GOTO 1460 ELSE PRINT:
PRINT:GOTO 1460

```

```

1460 LNO=LNO+LSP
1470 IF LNO<LMAX THEN RETURN
1480 GOSUB 1100      '>> EJECT PAGE
1490 RETURN
1500 '----- ERASE DOUBLING SP SUB. -----
1510 SSP=INSTR(INTXT$," ")
1520 SSP1=INSTR(SSP+1,INTXT$," "):IF SSP1=0 THEN RETURN
1530 IF SSP1-SSP=1 THEN INTXT%=LEFT$(INTXT$,SSP)+MID$(INTXT$,SSP1+1):GOTO 1520
ELSE SSP=SSP1:GOTO 1520
1540 '----- CENTERING SUB -----
1550 SHFT=(CMAX-LEN(PTXT$))/2:IF SHFT<=0 THEN RETURN
ELSE PTXT%=SPACE$(SHFT)+PTXT$:RETURN
1560 '----- TEXT LINE PROC. -----
1570 '>> MACRO TERM (CONSIDERING ESCAPE C)
1580 PSP=0      '>> 1-ST MACRO CHARA.
1590 PSP1=INSTR(PSP+1,INTXT$,ZMA$):IF PSP1=0 THEN 1690
1600 IF PSP1=1 THEN 1610 ELSE IF MID$(INTXT$,PSP1-1,1)=ZES$ THEN INTXT%=
LEFT$(INTXT$,PSP1-2)+MID$(INTXT$,PSP1):PSP=PSP1-1:GOTO 1590
1610 PSP=PSP1      '>> 2-ND MACRO CHARA.
1620 PSP2=INSTR(PSP1+1,INTXT$,ZMA$):IF PSP2=0 THEN 1690
1630 IF MID$(INTXT$,PSP2-1,1)=ZES$ THEN INTXT%=LEFT$(INTXT$,PSP2-2)+MID$(INTXT$,
PSP2):PSP=PSP2-1:GOTO 1620
1640 PSP=PSP2:MACT%=MID$(INTXT$,PSP1+1,PSP2-PSP1-1):IF MACT%="" THEN 1590
1650 FOR NZ=1 TO NM
1660 IF MACT%=MAC1$(NZ) THEN INTXT%=LEFT$(INTXT$,PSP1-1)+MAC2$(NZ)+MID$(INTXT$,
PSP2+1):GOTO 1590
1670 NEXT NZ:GOTO 1590
1680 '>> TEXT PROC. INTXT$--->PTXT$
1690 IF SFIL=0 THEN 1700 ELSE GOSUB 890:IF SCE=0 THEN PTXT%=INTXT$:EPL=1: GOSUB
1540:GOSUB 890:RETURN ELSE PTXT%=SPACE$(TAB)+INTXT$:EPL=1:GOSUB 890:RETURN
1700 '>> FILL TEXT CASE
1710 GOSUB 1500      '>> ERASE DOUBLING SP
1720 IF LEFT$(INTXT$,1)="" THEN GOSUB 890:PTXT%=SPACE$(TAB+CIN)+INTXT$: GOTO
1740
1730 IF EPL=0 THEN PTXT%=SPACE$(TAB)+INTXT$ ELSE PTXT%=PTXT%+" "+INTXT$
1740 EPL=1:LPTXT=LEN(PTXT$)
1750 IF LPTXT<CMAX THEN RETURN
1760 '>> SEARCH DELIMITING SP & DIVIDE TEXT LINE
1770 SSD=0
1780 SSD1=INSTR(SSD+1,PTXT$,"-"):IF SSD1<>0 AND SSD1<=CMAX THEN SSD=SSD1: GOTO
1780
1790 SSD=SSD+1
1800 SSP=0
1810 SSP1=INSTR(SSP+1,PTXT$," "):IF SSP1<>0 AND SSP1<=CMAX+1 THEN SSP=SSP1: GOTO
1810
1820 IF SSD>SSP THEN STXT%=MID$(PTXT$,SSD):PTXT%=LEFT$(PTXT$,SSD-1):GOSUB 890:
GOTO 1840
1830 STXT%=MID$(PTXT$,SSP+1):PTXT%=LEFT$(PTXT$,SSP-1):GOSUB 890
1840 PTXT%=SPACE$(TAB)+STXT$:GOTO 1740
1850 '=====
1860 END

```